
Abstract. Doaitse has always been enthusiastic about attribute gram-
mars, seeing them even where most people don’t. While I was writing up
my DPhil thesis, he explained to me where they were in that too. This
paper is by way of belated thanks for that perspective—and as a more
general rendering of the observation, with the benefit of hindsight and
twenty years of progress.

1 Introduction

I first met Doaitse Swierstra and became exposed to his predilections at the
STOP project summer school on Ameland in 1989. For as long as I have known
him, Doaitse has been devoted to attribute grammars—in compiler technology,
in program design, and simply in structuring his thought processes.

I next visited the Netherlands in the summer of 1991. I was deep in the middle
of writing up my DPhil thesis [1] on upwards and downwards accumulations on
trees, and during my trip I gave a couple of seminars about my work, including
one at Utrecht hosted by Doaitse. As many who have met him will attest, Doaitse
is like a Frenchman: whatever you say to him, he translates into his own language,
and forthwith it is something entirely different. In this case, he immediately
explained to me that my thesis was really about attribute grammars, and not
about accumulations at all.

In a sense, he was right. Indeed, his observation grew into the final chapter of
my thesis, which attempted to explain the connection. At the time of my visit, I
was looking for one more substantive piece of work to balance out what I already
had, so this contribution was very welcome. But I’m not sure that I ever thanked
him properly for his gift, and this is my opportunity to do so. Besides, I can now
tell the story much better than I could twenty-odd years ago: in particular, I can
do so datatype-generically.

So, thank you, Doaitse, for your ever fresh way of looking at the world!

2 Origami programming

One of the advances in functional programming since I did my thesis has been
a much better understanding of what I have been calling ‘datatype-generic’ [2]



and ‘origami’ [3] programming—that is, the use of structured recursion operators
such as maps, folds, and unfolds, parametrized by the shape of data. These ideas
were developing at the time, in the work of Hagino [4] and Malcolm [5], but I for
one didn’t really appreciate them until some years later. With the wisdom of
hindsight, the results in my thesis can be presented in this style too.

We will discuss attribute grammars in terms of labelling every node of a data
structure. So for simplicity, in this paper we stick to labelled data structures—
every node has precisely one label, of type a, and an f -structure of children:

data Mu f a = In {root :: a, kids :: f (Mu f a)}

The datatype Mu f supports numerous datatype-generic origami operations,
parametrized by the shape function f . Of these, we will only make use in this
paper of folds, not unfolds or other more sophisticated origami operators:

fold :: Functor f ⇒ (a → f b → b)→ Mu f a → b
fold φ t = φ (root t) (fmap (fold φ) (kids t))

We’ll also need ad-hoc datatype genericity; so we define a universe of polynomial
functors.

data Unit a = Unit
data Id a = Id a
data Sum f g a = Inl (f a) | Inr (g a)
data Prod f g a = f a :×: g a

Here also is the empty datatype—it will be useful for some constructions, al-
though we won’t consider it part of our universe of functors:

data Zero a

There are no constructors for Zero, so no (proper) values of that type; therefore,
if you were somehow to be able to come up a value of type Zero, you deserve to
be able to turn it into anything you want:

magic :: Zero a → b
magic z = seq z (error "It must be magic")

All these codes represent functors, of course:

instance Functor Zero where
fmap f z = magic z

instance Functor Unit where
fmap f Unit = Unit

instance Functor Id where
fmap f (Id a) = Id (f a)

instance (Functor f ,Functor g)⇒ Functor (Sum f g) where
fmap f (Inl x ) = Inl (fmap f x )
fmap f (Inr y) = Inr (fmap f y)

Fig. 1. The tree t = node 1 (leaf 2) (node 3 (leaf 4) (leaf 5))

instance (Functor f ,Functor g)⇒ Functor (Prod f g) where
fmap f (x :×: y) = fmap f x :×: fmap f y

For example, TreeF is the code for the shape functor of homogeneous binary
trees, and sum is an example fold for that datatype:

type TreeF = Sum Unit (Prod Id Id)
type Tree = Mu TreeF

add :: Int → TreeF Int → Int
add m (Inl Unit) = m
add m (Inr (Id n :×: Id p)) = m + n + p

sum :: Tree Int → Int
sum = fold add

For convenience, here are two ‘constructors’ for Tree:

leaf :: a → Tree a
leaf a = In a (Inl Unit)

node :: a → Tree a → Tree a → Tree a
node a t u = In a (Inr (Id t :×: Id u))

Figure 1 shows a small tree t , which we will use for examples.

3 Accumulations on lists

My thesis was about upwards and downwards accumulations on trees, which
were intended to be analogous to the accumulations (or ‘scans’) on lists that
had proven so fruitful in Bird’s work on the Theory of Lists [6–8]. Recall the
standard definitions of tails, foldr , and scanr from the Haskell libraries:

tails :: [a ]→ [[a ]]
tails [ ] = [[ ]]
tails x = x : tails (tail x )

foldr :: (a → b → b)→ b → [a ]→ b
foldr f e [ ] = e
foldr f e (a : x ) = f a (foldr f e x )

scanr :: (a → b → b)→ b → [a ]→ [b ]



scanr f e [ ] = [e ]
scanr f e (a : x ) = f a (head y) : y where y = scanr f e x

Here, scanr computes all the partial results of a fold, from the right:

scanr (+) 0 [1, 2, 3] = [6, 5, 3, 0]

It satisfies a very important ‘scan lemma’, stating that these partial results are
precisely the results of folding each of the tails:

scanr f e = map (foldr f e) ◦ tails

The scan lemma is crucial in deriving numerous efficient algorithms over lists,
not least for the famous ‘maximum segment sum’ problem [8].

Dually, there are functions that work from the opposite end of the list:

inits :: [a ]→ [[a ]]
inits [ ] = [[ ]]
inits (a : x ) = [ ] : map (a:) (inits x )

foldl :: (b → a → b)→ b → [a ]→ b
foldl f e [ ] = e
foldl f e (a : x ) = foldl f (f e a) x

scanl :: (b → a → b)→ b → [a ]→ [b ]
scanl f e [ ] = [e ]
scanl f e (a : x ) = e : scanl f (f e a) x

for which

scanl (+) 0 [1, 2, 3] = [0, 1, 3, 6]

Again, there is a scan lemma:

scanl f e = map (foldl f e) ◦ inits

With these six functions as my inspiration, the essence of my thesis work was to
generalize them to trees of various kinds.

4 Upwards accumulations

Generalizing tails and scanr is the easier task, because tail segments follow the
structure of the datatype, whereas inits and scanl in some sense go against the
grain. The idea is that tails labels every node of a list with the tail starting at
that position, and scanr labels every node with the fold of that tail. My thesis
made some ad-hoc generalizations of this idea to various kinds of tree. It wasn’t
until some years later [9] that a datatype-generic construction was discovered,
involving a systematic way of deriving a ‘labelled variant’ for any datatype, which
has (precisely) one label at each node. Happily, in this paper we have already
restricted ourselves to such datatypes, so we don’t need this construction—the
‘labelled variant’ of Mu f is Mu f itself.

Fig. 2. (a) subtrees t and (b) scanu add t

Then subtrees, the datatype-generic version of tails, labels every node of
a tree with the subtree rooted at that node. The root label of the output is
the whole of the input; and each child in the output is generated from the
corresponding child in the input.

subtrees :: Functor f ⇒ Mu f a → Mu f (Mu f a)
subtrees = fold ψ where ψ a ts = In (In a (fmap root ts)) ts

An upwards accumulation is like subtrees, except that it folds every tree it gen-
erates. It does this as it goes, so it takes no longer to compute than a mere fold
of the input tree does; which is to say, it records all the partial results already
involved in folding the original tree.

scanu :: Functor f ⇒ (a → f b → b)→ Mu f a → Mu f b
scanu φ = fold ψ where ψ a ts = In (φ a (fmap root ts)) ts

Note that we again have the all-important scan lemma:

scanu φ = fmap (fold φ) ◦ subtrees

Figure 2 shows (a) subtrees t and (b) scanu add t , where t is the tree in Figure 1.

5 Derivatives of datatypes

Generalizing inits and scanl is more difficult. The function inits labels every
node of a list with its list of predecessors, and the datatype-generic version
should label every node of a data structure with the ancestors of that node; but
the ancestors form a completely different datatype, of linear shape whatever the
branching structure of the original. Similarly, a downwards accumulation scand
will label every node of a data structure with some function of its ancestors—
and not just any function, but some kind of fold that will allow us to compute



the whole scan in linear time. To capture ancestors, we need to make a detour
through derivatives of datatypes [10, 11].

The derivative of a functor f is another functor ∆f such that ∆f a is like f a
but with precisely one a missing: ∆f a is a ‘one-hole context’ for f a with respect
to a. We model this idea through a type class Diff (of differentiable functors),
which has ∆ as an associated type synonym:

class Functor f ⇒ Diff f where
type∆f :: ∗ → ∗

The class also has two methods, to produce and to consume holes:

posns :: f a → f (a, ∆f a)
plug :: (a, ∆f a)→ f a

The idea is that posns takes a complete piece of data, and labels every element a
in it with the one-hole-context for which a completes the original data structure;
whereas plug takes a one-hole-context and a value to fill that hole, and puts the
latter in the former to make a complete piece of data. The two are related by the
following two laws (which I believe completely determine the implementation):

fmap fst (posns x ) = x
fmap plug (posns x ) = fmap (const x ) x

Informally, the first law states that posns really annotates, so that discarding
the annotations is a left inverse; and the second that plugging together each pair
in the output of posns produces many copies of the original data structure, one
for each element.

Here are the Diff instances for our universe of functors. The constant functor
has no elements, so the derivative is the empty type, posns has no effect, and
you can never have a hole to plug.

instance Diff Unit where
type∆Unit = Zero
posns Unit = Unit
plug (a, z ) = magic z

The identity functor contains precisely one element, so what’s left when this is
deleted is the unit type:

instance Diff Id where
type∆Id = Unit
posns (Id a) = Id (a,Unit)
plug (a,Unit) = Id a

A value of a sum type is either of the left summand or of the right, and in each
case a one-hole context is a corresponding one-hole context for that summand;
so the two methods simply follow the structure.

instance (Diff f ,Diff g)⇒ Diff (Sum f g) where
type∆(Sum f g) = Sum (∆f ) (∆g)



posns (Inl x ) = Inl (fmap (λ(a, dx )→ (a, Inl dx )) (posns x ))
posns (Inr y) = Inr (fmap (λ(a, dy)→ (a, Inr dy)) (posns y))
plug (a, Inl x ) = Inl (plug (a, x ))
plug (a, Inr y) = Inr (plug (a, y))

A value of a product type is a pair, and a one-hole context for the pair is either a
one-hole context for the left half, together with an intact right half, or an intact
left half and a context for the right half.

instance (Diff f ,Diff g)⇒ Diff (Prod f g) where
type∆(Prod f g) = Sum (Prod (∆f ) g) (Prod f (∆g))
posns (x :×: y) = (fmap (λ(a, dx )→ (a, Inl (dx :×: y))) (posns x )) :×:

(fmap (λ(a, dy)→ (a, Inr (x :×: dy))) (posns y))
plug (a, Inl (dx :×: y)) = plug (a, dx ) :×: y
plug (a, Inr (x :×: dy)) = x :×: plug (a, dy)

For example, consider the type TreeF Int , whose values are either a unit or a pair
of integers. The derivative ∆TreeF Int of this type represents data structures
with one missing integer. Expanding the definitions from the type class instances,
we see that

∆TreeF = Sum Zero (Sum (Prod Unit Id) (Prod Id Unit))

The left-hand variant of this sum is void: corresponding TreeF values are just a
unit, and there is no way for such a value to be missing an integer. The right-
hand variant is itself a sum: corresponding TreeF values have two integers, so
there are two ways for such a value to be missing an integer, and in each case
what remains is the other integer.

Here is a piece of data in the right-hand variant of TreeF Int :

u :: TreeF Int
u = Inr (Id 3 :×: Id 4)

Here are two one-hole contexts for u, in each case having the unit value in place
of one of the integers:

v1, v2 ::∆TreeF Int
v1 = Inr (Inl (Unit :×: Id 4))
v2 = Inr (Inr (Id 3 :×: Unit))

If you plug the correct integer back into each context:

u1, u2 :: TreeF Int
u1 = plug (3, v1)
u2 = plug (4, v2)

then you get the original data back again: u = u1 = u2.

5.1 Zippers

Incidentally, derivatives are intimately connected with zippers [12], which repre-
sent a data structure with a single subterm highlighted as a ‘focus’. Concretely,



a zipper is a pair. The first component is the subterm in focus. The second com-
ponent is the remainder of the data structure, expressed as a sequence of layers,
like an onion, innermost layer first; each layer is the one-hole context into which
the structure inside fits.

type Zipper f a = (Mu f a, [(a, ∆f (Mu f a))]) -- innermost layer first

To reconstruct the complete data structure from the zipper, we plug subterms
into contexts, from the inside out:

close :: Diff f ⇒ Zipper f a → Mu f a
close (x , ds) = foldl glue x ds where glue x (a, d) = In a (plug (x , d))

For example, consider a little tree t1 and two surrounding contexts tc2, tc3:

t1 :: Tree Int
t1 = leaf 4

tc2, tc3 :: (Int , ∆TreeF (Tree Int))
tc2 = (3, Inr (Inl (Unit :×: Id (leaf 5))))
tc3 = (1, Inr (Inr (Id (leaf 2) :×: Unit)))

The tree in Figure 1 can be reconstructed from these: t = close (t1, [tc2, tc3 ]).

6 Downwards accumulations

Now, the ancestors of an element in a data structure form a path to that element,
and paths are a projection of zippers. They omit the subterm in focus; they also
omit all non-ancestors (siblings, great-aunts, etc) of the focus too, so the type
parameter to ∆f is the unit type. We call such values ‘directions’, because they
state which direction to take in a parent to get to one of its children.

type Dir f = ∆f ()

For example, the derivative of TreeF is a sum type, and so there are different
possible directions for each variant. However, as we have already seen, the left-
hand summand of ∆TreeF is Zero, indicating that there is no direction you can
turn to get to a child of a leaf. The right-hand summand is another sum type,
and this yields two possible directions to turn for a child of an internal node.

left , right :: Dir TreeF
left = Inr (Inl (Unit :×: Id ()))
right = Inr (Inr (Id () :×: Unit))

Each node of a Mu f a data structure has a label of type a and an f -structure
of children; so a path to a node (including the root label of the target node)
consists of an alternating sequence of elements a and directions Dir f , with one
more element than direction. We represent this sequence innermost-first, so that
the path to a child has as a subterm the path to its parent.

data Path f a = Start a | Step (Path f a) (Dir f ) a



For example, the path to the node labelled 4 in t starts with a 1, turns right
to meet a 3, then turns left to meet a 4, so it is represented by the expression
Step (Step (Start 1) right 3) left 4.

There is, of course, a natural pattern of folds for paths:

foldPath :: (b → Dir f → a → b)→ (a → b)→ Path f a → b
foldPath f g (Step p d a) = f (foldPath f g p) d a
foldPath f g (Start a) = g a

The function paths takes a data structure and labels every node with the
path from the root to that node:

paths :: (Diff f ,Functor (∆f ))⇒ Mu f a → Mu f (Path f a)

The definition will have to involve an accumulating parameter: the paths to
children depend on surrounding context (the ‘path so far’) as well as the children
themselves. It seems sweetly reasonable to start by unpacking the data structure
In a ts using posns, labelling every child t in ts with its one-hole context. We
therefore require an auxilliary function paths ′ that should depend on a path so
far, initially simply Start a, and apply to both a child and its one-hole context:

paths (In a ts) = In p (fmap (paths ′ p) (posns ts)) where p = Start a

paths ′ :: (Diff f ,Functor (∆f ))⇒
Path f a → (Mu f a, ∆f (Mu f a))→ Mu f (Path f a)

From here, the remainder of the definition is basically driven by the types. We
turn the one-hole context z into a direction by discarding siblings, and use that
and the node label to extend the path so far for recursive calls.

paths ′ p (In a ts, z ) = In q (fmap (paths ′ q) (posns ts))
where q = Step p (fmap bang z ) a

Here, bang is a basic combinator for the unit type:

bang :: a → ()
bang a = ()

Figure 3(a) shows paths t , where t is the tree in Figure 1.
A downwards accumulation is then a fold mapped over the paths:

scand :: (Diff f ,Functor (∆f ))⇒
(b → Dir f → a → b)→ (a → b)→ Mu f a → Mu f b

scand f g = fmap (foldPath f g) ◦ paths

But because we carefully arranged that the paths to children share a subterm
with the path to their parent, we can compute this incrementally in linear time,
assuming that the basic operations take constant time. Again, we use an accu-
mulating parameter; but this time, it will be the image under foldPath f g of the
path so far rather than the path itself.

scand f g (In a ts) = In b (fmap (scand ′ f g b) (posns ts)) where b = g a

Fig. 3. (a) paths t and (b) routes t (the latter edited for presentation)

As with paths, the remainder of the definition is type-driven: we turn the one-
hole context into a direction by discarding siblings, and use that and the node
label to update the accumulating parameter for recursive calls.

scand ′ :: (Diff f ,Functor (∆f ))⇒
(b → Dir f → a → b)→ (a → b)→ b → (Mu f a, ∆f (Mu f a))→ Mu f b

scand ′ f g b (In a ts, z ) = In c (fmap (scand ′ f g c) (posns ts))
where c = f b (fmap bang z ) a

(In fact, the second argument g of scand ′ is not used.)
For example, we can produce a little guidebook for a tree, recording in user-

friendly format the route to each node in the tree:

routes :: Show a ⇒ Tree a → Tree String
routes = scand step start

start a = "Start, " ++ show a
step s (Inl z ) a = magic z
step s (Inr (Inl (Unit :×: Id ()))) a = s ++ "; Left, " ++ show a
step s (Inr (Inr (Id () :×: Unit))) a = s ++ "; Right, " ++ show a

The first clause for step is not really needed, because it can never be called;
but its presence makes the definition manifestly total. The results are shown in
Figure 3(b) (with the strings edited for presentation).

7 Attribute grammars

Having seen datatype-generic definitions of upwards and downwards accumula-
tions, let us now return to the topic of attribute grammars. These were proposed
by Knuth [13] as a tool for presenting the semantics of programming languages.
They arose as an extension of the ‘syntax-directed’ compilation techniques of
Irons and others in the early sixties [14]. Using these techniques, the parse tree



of a program is decorated with attributes, the decoration attached to an element
of the parse tree representing some aspect of the semantics of the subtree rooted
there. In Irons’ formulation, the attribute attached to an element depends only
on the descendants of that element; Knuth showed that although no extra power
is gained by doing so, the description of the semantics of a language can be con-
siderably simplified by allowing attributes to depend on other parts of the parse
tree as well.

Traditionally, an attribute grammar for a context free language is an exten-
sion of the grammar which describes the syntax of that language. Each symbol
in the grammar is associated with a number of attributes, and each produc-
tion in the grammar comes with some rules that give values to some of the
attributes attached to symbols appearing in that production, in terms of the
values of the other attributes that appear. The attributes are classified into two
categories, inherited and synthesized ; inherited attributes are those appearing
on the right hand side of the production in which their value is defined, and
hence concern the ‘children’ of the production, whereas synthesized attributes
appear on the left, and concern the parents. Irons’ syntax-directed translation
corresponds to attribute grammars with only synthesized attributes. Intuitively,
inherited attributes carry information into a subtree and synthesized attributes
carry it back out again; in Knuth’s [15] words, ‘inherited attributes are, roughly
speaking, those aspects of meaning which come from the context of a phrase,
while synthesized attributes are those aspects which are built up from within
the phrase.’

Our view of attribute grammars differs somewhat from this traditional view,
and follows instead the approach pioneered by Doaitse and his colleagues [16,
17]. We suppose that a tree has been built already, and that the task is simply to
evaluate the attributes. That is, attribute grammars are viewed as a means for
describing computations over pre-existing trees, rather than in the ‘grammar’
sense for recognizing or generating those trees in the first place.

We make the simplification, after [18], that every element has exactly one
inherited and one synthesized attribute, and that all inherited attributes have
the same type i , and all synthesized attributes the same type s. This entails no
loss of generality, since attribute types may be product types.

In our datatype-generic formulation, there is just one datatype construc-
tor In, so a single production rule suffices. The production rule takes as input
the label (of type a) of a node, the value (of type i) of the sole inherited attribute
for that node, and values (collectively of type f s) for the synthesized attributes
of each of the children. It should yield as output the value (of type s) of the
sole synthesized attribute of the node, and inherited attributes (collectively of
type f i) for each of the children. We capture this via the following type synonym:

type Rule f a i s = a → (f s, i)→ (s, f i)

We call a rule r :: Rule f a i s ‘shape-preserving’ if the inherited attributes it
generates always match up with the children; that is, for appropriate inputs
a, ss, i , if (s, is) = r a (ss, i) then fmap bang is = fmap bang ss. We restrict
attention in this paper to shape-preserving rules. Of course, ‘matching up’ will



require the ability to zip together f -structures; we will declare a suitable type
class Zippable with member function zip shortly.

7.1 Lazy evaluation and cyclic programs

The usual formulation of attribute grammars is then to compute the synthesized
attribute of the root node of a term, given the inherited attribute. Convention-
ally a lot of effort goes into deducing the dataflow constraints that arise; for
example, maybe an inherited attribute of a right child depends on a synthesized
attribute of a left sibling, and so a left-to-right traversal is called for. However,
as Johnsson [19] observed, in a lazily evaluated language this can all be ignored:
the evaluation mechanism automatically works out the appropriate dataflow. So
attribute evaluation can be captured as a simple cyclic lazy functional program:

eval :: (Functor f ,Zippable f )⇒ Rule f a i s → (Mu f a, i)→ s
eval r (In a ts, i) = s
where

(s, is) = r a (ss, i)
ss = fmap (eval r) (zip ts is)

Observe that the inherited attributes is of the children and their synthesized
attributes ss are defined using mutual recursion. Statically checking that this
recursion is well founded is inherently exponential [20]; indeed, it was one of
the first naturally occurring problems to be shown so. In this paper, we assume
well-foundedness.

7.2 Matching up

However, one clear necessary condition for well-foundedness is for the shape of
zip ts is to be determined purely by the shape of ts, independent of is:

fmap bang (zip ts is) = fmap bang (zip (fmap bang ts)⊥)

This is no hardship, because a given ts can be zipped successfully with only one
shape of is. Operationally, we require a datatype-generic zip that is non-strict
in its second argument. We introduce a type class of zippable functors:

class Zippable f where
zip :: f a → f b → f (a, b)
select :: f a → ∆f ()→ a

We have added an operation select too, which takes an f -structure of elements
and a position in that structure, and selects the appropriate element; we will use
this later. All functors in our universe are zippable:

instance Zippable Unit where
zip Unit unit = let Unit = unit in Unit
select Unit z = magic z

instance Zippable Id where



zip (Id a) id b = let Id b = id b in Id (a, b)
select (Id a) Unit = a

instance (Zippable f ,Zippable g)⇒ Zippable (Sum f g) where
zip (Inl x ) inl y = let Inl y = inl y in Inl (zip x y)
zip (Inr x ) inr y = let Inr y = inr y in Inr (zip x y)
select (Inl x ) (Inl d) = select x d
select (Inr y) (Inr d) = select y d

instance (Zippable f ,Zippable g)⇒ Zippable (Prod f g) where
zip (x :×: y) prod x ′ y ′ = let x ′ :×: y ′ = prod x ′ y ′ in

zip x x ′ :×: zip y y ′

select (x :×: y) (Inl (d :×: y ′)) = select x d
select (x :×: y) (Inr (x ′ :×: d)) = select y d

Note the asymmetry in the definitions of zip, to ensure non-strictness in the
second argument. Of course, zip and select are partial functions, because the
shapes might not match—specifically in the Sum case. To be more precise about
typing, we could make both methods return a Maybe result. However, if we
stick to shape-preserving attribute rules, then we will only ever use these two
functions on matching shapes.

7.3 Attribute evaluation as a fold

The definition of eval above is not obviously in a structured form, since it uses
explicit recursion. However, the only use of ts is in recursive calls, so it isn’t
difficult to rearrange the definition into a fold. Of course, children should be
processed using different values for the inherited attributes, so it isn’t eval itself
that is a fold. In fact, as many have observed [21–23, 19, 18], it is curry eval ,
computing from a tree a function of type i → s, that is the fold.

curryeval :: (Functor f ,Zippable f )⇒ Rule f a i s → Mu f a → i → s
curryeval r = fold φ
where φ a fs i = let (s, is) = r a (ss, i)

ss = fmap (uncurry ($)) (zip fs is)
in s

Conversely, any fold can be formulated as an attribute grammar using only syn-
thesized attributes. From the algebra φ for a fold, we can construct an attribute
grammar production rule upRule φ such that curryeval (upRule φ) () = fold φ,
using the unit type for inherited attributes:

upRule :: Functor f ⇒ (a → f b → b)→ Rule f a () b
upRule φ a (bs, ()) = (φ a bs, fmap bang bs)

7.4 Complete attribute evaluation

Attribute evaluation is conventionally understood to mean evaluation of a single
attribute, the synthesized attribute of the root of the tree; all the other attributes



are ‘intermediate results’ and are of no further interest. For most applications,
and in particular for one-off compilation, this is exactly what is required; once
the translation of part of a program has been constructed, the translations of
subprograms are no longer needed. However, for some applications we want the
intermediate results as well; for example, incremental compilers and structure
editors such as the Cornell Synthesizer Generator [24] make use of these inter-
mediate results in order to avoid having to recompile parts of a program that
remain unchanged. For such applications, we would like attribute evaluation to
return the whole tree of attributes, not just the synthesized attribute of the root.

We have seen that the curried evaluation function curryeval is a fold; so
fmap curryeval ◦ subtrees, yielding a tree of inherited-to-synthesized-attribute
functions, is an upwards accumulation. This is not quite enough to allow us to
compute all the attributes in the tree: given the inherited attribute of the root,
we can certainly find the synthesized attribute of the root, but what will the
inherited attributes of the children be? We have thrown that information away.

In fact, to support incremental attribute re-evaluation, we should annotate
the input tree to record the values of both the inherited and the synthesized at-
tributes at each node. That can be achieved by a slight modification to curryeval ,
reconstructing the input tree as we go, but retaining the inherited attributes
throughout. (For completeness, we also preserve the original labels.)

annotate :: (Functor f ,Zippable f )⇒ Rule f a i s → Mu f a → i → Mu f (a, i , s)
annotate r = fold (ψ r) where
ψ r a fs i = In (a, i , s) ts where

(s, is) = r a (ss, i)
ts = fmap (uncurry ($)) (zip fs is)
ss = fmap (thd3 ◦ root) ts

(This is roughly the concluding construction in [25].)

An illuminating example of an attribute grammar making essential use of
both inherited and synthesized attributes is the ranking problem, in which each
node is labelled with its position in left-to-right order. This can be expressed as
an attribute grammar with one inherited attribute, representing the first index
to use, and two synthesized attributes, recording for each node the rank of that
node and the size of (that is, the number of elements in) the subtree rooted
at that node. For a binary tree, the inherited attribute passed in to a node is
propagated to the left child; but the inherited attribute passed to the right child
depends also on the size of the left child. Information flows from left to right,
in the same way that it does for a depth-first search. Most of the applications
of attribute grammars to programming languages involve dependencies like this,
because of the close correspondence between the hierarchical structure of the
parse tree and the linear structure of the program text it represents. For our
binary tree datatype, we have the following rule:

rankRule :: Rule TreeF a Int (Int , Int)
rankRule a (Inl Unit , i) = ((i , 1), Inl Unit)



rankRule a (Inr (Id (r1, s1) :×: Id (r2, s2)), i)
= ((i+s1, 1+s1+s2), Inr (Id i :×: Id (i+s1+1)))

Then rank ordering is computed by complete attribute evaluation according to
this rule, followed by discarding the auxilliary data (the starting index and the
size):

rank :: Tree a → Tree (a, Int)
rank t = fmap (λ(a, i , (r , s))→ (a, r)) (annotate rankRule t 0)

As another example, consider Bird’s ‘repmin’ problem [26]: replace every el-
ement of a tree with the minimum element in that tree. Bird shows a clever
circular one-pass solution; as was pointed out soon afterwards, not least (and
perhaps first?) by Doaitse himself [27, 19], this circular program precisely cor-
responds to lazy evaluation of an attribute grammar. Here, the sole inherited
attribute is the value with which to replace all labels, and the sole synthesized
attribute for a node is the minimum value in the subtree rooted there:

repminRule :: Ord a ⇒ Rule TreeF a a a
repminRule a (Inl Unit ,m) = (a, Inl Unit)
repminRule a (Inr (Id x :×: Id y),m)

= (min a (min x y), Inr (Id m :×: Id m))

Solving the repmin problem in a single pass then amounts to complete attribute
evaluation according to this rule, followed by discarding the original labels and
the synthesized attributes—but ensuring that the initial value of the inherited
attribute is the synthesized attribute associated with the root:

repmin :: Ord a ⇒ Tree a → Tree a
repmin t = let u = annotate repminRule t (thd3 (root u)) in fmap snd3 u

(where snd3 (a, b, c) = b and thd3 (a, b, c) = c). The grammar rule is acylic; it
is this main program that ties the cyclic knot—u is defined in terms of itself.

7.5 Accumulations as attribute evaluation

Now, complete attribute evaluation in the sense of the previous section general-
izes both upwards and downwards accumulations. As we have seen, the algebra
for a fold can be expressed as an attribute grammar rule using only synthe-
sized attributes, and evaluating the fold amounts to computing the synthesized
attribute of the root node using that rule. As one might therefore expect, any up-
wards accumulation can be computed by a complete attribute evaluation using
the same attribute rule (and then discarding the original labels and the trivial
inherited attribute values):

scanu φ t = fmap thd3 (annotate (upRule φ) t ())

Dually, any downwards accumulation can be computed by a complete at-
tribute evaluation using only inherited attributes, using the following rule:



downRule :: (Diff f ,Functor (∆f ))⇒
(b → Dir f → a → b)→ (a → b)→ Rule f a (a → b) ()

downRule f g a (ss, k) = ((), fmap (f (k a) ◦ fmap bang ◦ snd) (posns ss))

Note that the inherited attribute for a node can depend only on context outside
the tree rooted there, so in particular cannot depend on the node label. Therefore
we make the inherited attribute a function from node label to result, and finish
off the computation by applying the inherited attributes to the original labels:

scand f g t = fmap (λ(a, i , ())→ i a) (annotate (downRule f g) t g)

7.6 Attribute evaluations as accumulation

Clearly, there is a strong analogy between complete attribute evaluation and
upwards and downwards accumulations: an upwards accumulation is the com-
plete evaluation of an attribute grammar with only synthesized attributes, and
a downwards accumulation is the complete evaluation of a grammar with only
inherited attributes.

Conversely, it turns out that any complete attribute evaluation can be ex-
pressed as an upwards accumulation followed by a downwards accumulation. The
idea is to make a first pass over the tree, labelling every node with (the original
node label and) a function from the input inherited attribute to the synthesized
attribute of that node together with the inherited attributes for the children;
then to finish up by composing all the inherited-to-inherited-attribute functions
along each of the paths from the root. The first pass is an upwards accumulation:

collect :: (Functor f ,Zippable f )⇒
Rule f a i s → Mu f a → Mu f (a, i → (s, f i))

collect r = scanu (λa afs → (a, φ r a afs)) where
φ r a afs i = (s, is ′) where

(s, is) = r a (ss, i)
fis = zip (fmap snd afs) is
ss = fmap (fst ◦ uncurry ($)) fis
is ′ = fmap snd fis

The second is a downwards accumulation, followed by a map to discard the
children’s inherited attributes:

distribute :: (Diff f ,Functor (∆f ),Zippable f )⇒
i → Mu f (a, i → (s, f i))→ Mu f (a, i , s)

distribute i t = fmap tidy (scand step (start i) t) where
step ( , , ( , is)) d (a, h) = let i = select is d in (a, i , h i)
start i (a, h) = (a, i , h i)
tidy (a, i , (s, is)) = (a, i , s)

Putting them together, we have

annotate r t i = distribute i (collect r t)



Incidentally, this pattern of ‘collect information upwards through the tree, then
redistribute it downwards’ is very common; in my thesis [1] and subsequent
papers [28, 29] I showed that it also crops up in efficient parallel algorithms for
computing prefix sums and in drawing trees.

